Derivatives exposure with Mathematica 10

Financial derivatives such as swaps or forward rate agreements have a tendency to change their value over the life of the contract. This means they can become an asset of a liability to a contractual party. When contract is as asset and has a positive mark-to-market value, it creates an exposure to counterparty. Such exposure is a risk factor (known as counterparty exposure) since it can be lost if the counterparty gets into financial distress. Therefore, exposures are dully controlled and monitored as part of counterparty risk management.
Future exposures are calculated either analytically or through the Monte Carlo simulation and key is the understanding of the future value of the contract. **Mathematica 10** comes as excellent choice for this task as it offers enormous flexibility to handle various exposures calculation either way.

Let's assume an institution has entered into 5 years 2% interest rate swap on the nominal N and wants to establish the future exposure of this contract. To calculate the exposure, the institution needs to know: (i) the current value of the contract and (ii) its future paths. Whilst (i) is straightforward, the (ii) is not since the future value is unknown and has to be estimated. In order to estimate the future value, one needs to know how this future value can evolve – i.e. probabilistic interpretation of unknown outcome.

Contract value: $V = (F - S) * Duration$ where F is the future swap rate and S is the transacted swap rate.
Future swap rate [F]: let's assume the swap rate follows Geometric Brownian Motion – a logarithmic SDE process with a drift μ =2% and volatility $\sigma = 20\%$. We apply the MC simulation to determine the future exposure:

```
gbproc = GeometricBrownianMotionProcess[0.02, 0.2, 0.02];
procsim = RandomFunction[gbproc, {0, 5, 0.01}, 300];
ListLinePlot[procsim["Path", Range[15]],
 PlotLabel → "Simulated paths of the swap rate"]
```



Once we know the future swap rate evolution, we can compute the future contract value.

```
SV[T_, S_, t_, v_] := (T - t) * (v - S);

tsm = TimeSeriesMapThread[SV[5, 0.02, #1, #2] &, procsim];
ListLinePlot[tsm["Path", Range[15]],
 PlotLabel → "Simulated future contract value"]
```

And we can visualise it as follows:



The future contract value can be both positive and negative – if the future swap rate F stays above 2%, the contract value will be positive, otherwise it will become negative.

It is also worth noting the profile of the future contract value – it is zero at the beginning and the end, despite the fact that future swap rate can be very different from the transacted rate. This is due to the duration effect that shortens with the passing time and reaches zero when the contact matures.

Exposure arises only when the mark-to-market value of the contract is positive. This quantity is also known as **Potential Future Exposure** [PFE]
We can define a new function that handles the PFE calculation: $PFE(t) = Max[V(t), 0]$

```
Exposure[T_, S_, t_, v_] := (T - t) * Max[v - S, 0];
epesim = TimeSeriesMapThread[Exposure[5, 0.02, #1, #2] &, procsim];
ListLinePlot[epesim["Path", Range[15]], PlotRange → All,
 PlotLabel → "Future exposure"]
```

This is the future exposure profile



which is the positive part of the previous graph.

Knowing the distribution profile of the PFE, we can compute various useful risks associated with the exposure mathematics:

Peak exposure [PE] – this is the PFE calculated at certain confidence level – say 95%

```
PE = TimeSeriesThread[Quantile[#, 0.95] &, epesim];
ListLinePlot[PE, PlotStyle → Blue,
 PlotLabel → "Peak exposure profile"]
```



Maximum peak exposure[MPE] – this is the maximum value of the PE:

```
peak = FindPeaks[PE, 0, 0, Max[PE]]["Path"]
{{1.84, 0.0364758}}
ListPlot[{PE, peak}, Joined → {True, False},
 PlotStyle → {Blue, {Red, PointSize[0.03]}},
 Filling → {2 → 0}, Ticks → {Automatic, True}]
```



The peak exposure of 3.65% of the swap nominal is projected to occur at 1.84 years from now.

Expected exposure [EE] – this is additional useful measure frequently used in the risk management and provides the view on the average or expected exposure over the life of the swap:

$EE(t) = E[Max[V(t), 0]]$

```
EE = TimeSeriesThread[Mean, epesim];
ListLinePlot[EE, PlotStyle → Magenta,
 PlotLabel → "Expected exposure"]
```



Although the chart profile looks the same, the EE is quite different numerically when compared to the maximum or 95% peak exposures:



Expected positive exposure [EPE] – this is the time-weighted EE over the life of the contract:

$$EPE(t) = \frac{1}{T}\int EE(t)dt$$

```
ListLinePlot[Accumulate[EE["Values"] * 0.01] / EE["Times"],
  PlotStyle → {Red, Thick},
  PlotLabel → "Expected postive exposure"] // Quiet
```



This quantity is generally used in the CVA calculation – the credit valuation adjustment of the contract due to the possibility of counterparty default.

Effective expected exposure [EEE] – this is the maximum EE over the time horizon:

$EEE(t) = Max[EE(t)] \quad 0 \leq s \leq t$

```
EEE = MovingMap[Max, EE, 30];
ListPlot[{EEE, peak2}, Joined → {True, False},
 PlotStyle → {Purple, {Red, PointSize[0.03]}},
 Filling → {2 → 0}, Ticks → {Automatic, True},
 PlotLabel → "Effective expected exposure with peak value"]
```



The max exposure = 0.083% at 1.5 years from now.